ANGULAR FUNCTIONS BY ELECTRONIC COMPUTER

By

JERRY GENE WILLIAMS

Bachelor of Science

Oklahoma State University

Stillwater, Oklahoma

1961

Submitted to the Faculty of the Graduate School of
the Oklahoma State University
in partial fulfillment of the requirements
for the degree of
MASTER OF SCIENCE
August, 1962

ANGULAR FUNCTIONS BY

ELECTRONIC COMPUTER

THESIS APPROVED:

J M Gillespie
Thesis Adviser

Dean of Graduate School

ACKNOWLEDGEMENTS

For making this graduate study possible, the writer wishes to express his indebtedness and appreciation to the following individuals and organizations:

To God for giving the writer a sound mind and body with which to pursue knowledge;

To the Continental Oil Company and the School of Civil Engineering for awarding the fellowship which made this graduate study financially possible;

To Dr. James W. Gillespie for his guidance and inspiration both as an instructor and as the writer's adviser;

To the faculty of the School of Civil Engineering and related departments, both for providing the undergraduate background to pursue graduate study and for their aid and instruction during the graduate study;

To Professor Granet and the staff of the Oklahoma State University computing center for making the facilities of the center available;

To his parents for their love, understanding, and sacrifice from the very beginning of his education.

In addition, gratitude is due Mrs. Ann Reichman who typed the final manuscript.

TABLE OF CONTENTS

LIST OF TABLES

Table Page

LIST OF ILLUSTRATIONS

Figure Page

NOMENCLATURE

A - cross sectional area

b - width of plate

E - modulus of elasticity

F_{AB}, F_{BA} - angular flexibility

G_{AB}, G_{BA} - angular carry-over

h - depth of plate

I - moment of inertia

L - length of span

L_i, L_j, L_k - length of segments i, j, and k

M_i, M_j, M_k - bending moments at points i, j, and k

m - number of plates

n - number of segments

P - intensity of concentrated load

$\bar{P}$ - elastic weight

S - section modulus

w - intensity of load

DL - dead load

UL - uniform load

Δ - deflection

ρ - density of construction material

τ_{ij}, τ_{ji} - angular load functions

CHAPTER I

INTRODUCTION

1-1 Discussion

The analysis of a continuous plate girder structure may be solved by either the flexibility or stiffness approach. In this thesis the flexibility method is considered with the three moment equation adopted as the mathematical model for analysis.

Computing angular functions for a structure which possesses numerous cross sectional variations, by conventional methods, is a laborious process. Much current literature is advocating the use of electronic computers for structural problems which are of a repetitive nature. The primary benefits obtained are economy, increase in productivity, solutions with greater accuracy, and optimization of design.

Many highway departments have developed computer programs to assist in the analysis and design of simple and continuous bridges. A series of four reports on the analysis of continuous beam bridges published by the School of Civil Engineering at Oklahoma State University (4, 7, 8, 10) is concerned with the complete analysis of a parabolic haunched bridge by electronic computer. Exline (3) presented a computer program for angular functions for a member with abrupt change in cross section but limited to members with a maximum of three variations.

In this thesis three separate programs for the IBM 650 electronic computer are developed in FORTRAN language utilizing floating point arithmetic. These programs are for sectional properties, angular functions, and deflections due to dead or uniform loading.

1-2 Flexibility Method

Having calculated the angular flexibilities, angular carry-over functions, and angular load functions for a structure of variable cross section by electronic computer, the three moment equation flexibility matrix is obtained by substitution. By inverting the flexibility matrix, a solution for final end moments may be obtained.

The three moment equation

$$G_{ij} M_i + \Sigma F_j M_j + G_{kj} M_k = -\Sigma \tau_j$$

may conveninetly be written for each redundant and placed in matrix form.

$$\begin{bmatrix} \Sigma F_o & G_{1o} & & & & & \\ G_{o1} & \Sigma F_1 & G_{21} & & & & \\ & \cdot & \cdot & \cdot & & & \\ & & G_{ij} & \Sigma F_j & G_{kj} & & \\ & & & \cdot & \cdot & \cdot & \\ & & & & G_{n-2,n-1} & \Sigma F_{n-1} & G_{n,n-1} \\ & & & & & G_{n-1,n} & \Sigma F_n \end{bmatrix} \begin{bmatrix} M_o \\ M_1 \\ \cdot \\ M_j \\ \cdot \\ M_{n-1} \\ M_n \end{bmatrix} = (-1) \begin{bmatrix} \Sigma \tau_o \\ \Sigma \tau_1 \\ \cdot \\ \Sigma \tau_j \\ \cdot \\ \Sigma \tau_{n-1} \\ \Sigma \tau_n \end{bmatrix} \qquad (1)$$

Using abbreviated notation, equation (1) can be rewritten

$$\begin{bmatrix} F_j \end{bmatrix} \left\{ M_j \right\} = (-1) \left\{ \Sigma \tau_j \right\} \qquad (2)$$

where:

$\left[F_j\right]$ - flexibility matrix

Inversion of the flexibility matrix yields a solution for final end moments.

$$\left\{M_j\right\} = -\left[F_j\right]^{-1} \left\{\Sigma\, \tau_j\right\} \tag{3}$$

Programs for matrix inversions are readily available; thus, they have not been developed in this thesis. Once final end moments are known, final deflections may be obtained as developed in Chapter IV.

CHAPTER II

CROSS SECTIONAL PROPERTIES

2-1 Discussion

Determination of the sectional properties of each unique cross section along the span is necessary for the evaluation of angular functions. Those properties considered and programed include:

A_x - cross-sectional area at section x

I_x - moment of inertia at section x

h_x - total depth at section x

S_{Tx} - section modulus for top at section x

S_{Bx} - section modulus for bottom at section x

The moment of inertia I_x is used directly in the evaluation of angular functions (Chapter III). The cross sectional area A_x is needed in the evaluation of angular functions for dead load. Although the application of the sectional moduli S_{Tx} and S_{Bx} is beyond the scope of this thesis, they are easily obtained and are included for use in the possible extension of this thesis. The total depth of each section is also included for the designer's convenience.

2-2 Derivation

A typical span AB (Fig. 2-1) of a continuous plate-girder of variable cross section is divided into n segments. For segments



Figure 2-1

Typical Span AB of n Unequal Strips



Figure 2-2

Typical Cross Section With Centroid Location of uniform cross section each change in cross section may be considered the termination and beginning of a segment. The width and depth of a typical plate at section x is designated as b_{xj} and h_{xj} respectively (Fig. 2-2). The width and depth of a segment of non-uniform cross section must be approximated by some appropriate method. For example, the depth of a segment whose depth is not constant may be approximated as the depth at mid-segment or as an average of the end depths. Numbering of sections begins with two to more easily facilitate writing of the computer program for Phase II (Angular Functions). Each time a subscript designates a particular segment, the cross section to the left of the subscript is inferred.

The cross sectional area of segment x is

$$A_x = \sum_{j=1}^{m} b_{xj}\ h_{xj} \tag{2-1}$$

The distance from the bottom plate to the centroid of the section (Fig. 2-2) may be evaluated as

$$\overline{Y}_{Bx} = \frac{\sum_{j=1}^{m} (b_{xj}\ h_{xj})(\overline{Y}_{Bx,j})}{\sum_{j=1}^{m} b_{xj}\ h_{xj}} \tag{2-2}$$

The distance from the top plate to the centroid of the section is

$$\overline{Y}_{Tx} = \sum_{j=1}^{m} (h_{xj}) - \overline{Y}_{Bx} \tag{2-3}$$

The distance from the centroid of the section to the centroid of each plate is

$$\bar{Y}_{xj} = \bar{Y}_{Bx} - \bar{Y}_{Bx,j} \tag{2-4}$$

The moment of inertia of a typical section is

$$I_x = \sum_{j=1}^{m} \left[\frac{b_{xj}\, h_{xj}^{\,3}}{12} + b_{xj}\, h_{xj}\, \bar{Y}_{xj}^{\,2} \right] \tag{2-5}$$

And the corresponding section moduli are

$$S_{Tx} = \frac{I_x}{\bar{Y}_{Tx}}$$

and (2-6)

$$S_{Bx} = \frac{I_x}{\bar{Y}_{Bx}}$$

2-3 Computer Program

2-3-1 Results of Program

The program presented (Fig's. 2-3,4) is written for the IBM 650 electronic computer using FORTRAN language and floating point arithmetic. Notation of typical section x in the derivation has been changed to section i in the computer program. The program yields a solution for:

AREA(I) - the cross sectional area of section i

EYE(I) - the moment of inertia of section i

SXTOP(I) - the section modulus for the top for section i

SXBOT(I) - the section modulus for the bottom for section i

DEPTH(I) - the total depth of section i

2-3-2 Data Required

The input data necessary for computing the desired sectional properties include:

N - the total number of strip segments

M(I) - the total number of plates at section i

B(I,J) - the width of plate j at section i

H(I,J) - the depth of plate j at section i

2-3-3 Auxiliary Quantities

Other quantities integral in the program are denoted as:

YBOT(I) - the distance from the bottom to the centroid of section i

YTOP(I) - the distance from the top to the centroid of section i

YBAR(I,J) - the distance from the bottom of section i to the centroid of plate j

YPLAT(I,J) - the distance from the centroid of section i to the centroid of plate j



Figure 2-3

Flow Chart Phase I - Sectional Properties

```
C 0000 0 PHASE ONE-EVALUATION OF CROSS
C 0000 0 SECTIONAL PROPERTIES  AREA,
C 0000 0 MOMENT OF INERTIA,DEPTH,AND
C 0000 0 SECTION MODULI
     1 0 DIMENSION AREA(25), EYE(25),
     1 1 YBOT(25),YTOP(25), SXTOP(25),
     1 2 SXBOT(25), M(25), B(25,9)
   100 0 DIMENSION H(25,9), YBAR(25,9),
   100 1 YPLAT(25,9), DEPTH(25),
   100 2 TYBOT(25)
     2 0 READ, N
     3 0 DO 103 I=2,N
     4 0 READ, M(I)
   101 0 L = M(I)
   102 0 DO 103 J=1,L
   103 0 READ, B(I,J), H(I,J)
C 0000 0 CALCULATE CROSS SECTIONAL AREA
     5 0 DO 9 I=2,N
     6 0 AREA (I) = 0.0
     7 0 L = M(I)
     8 0 DO 9 J=1,L
     9 0 AREA(I) = AREA(I) + B(I,J)*
     9 1 H(I,J)
C 0000 0 CALCULATE DISTANCE FROM
C 0000 1 BOTTOM TO CENTROID OF
C 0000 2 EACH PLATE - YBAR
    10 0 DO 14 I=2,N
    11 0 YBAR(I,1) = H(I,1)/2.0
    12 0 L = M(I)
    13 0 DO 14 J=2,L
    14 0 YBAR(I,J) = YBAR(I,J-1)
    14 1 + H(I,J-1)/2.0 + H(I,J)/2.0
C 0000 0 CALCULATE DISTANCE FROM
C 0000 0 BOTTOM TO CENTROID OF SECTION
    15 0 DO 16 I =2,N
    16 0 TYBOT(I) = 0.0
    17 0 DO 21 I=2,N
    18 0 L =M(I)
    19 0 DO 20 J=1,L
    20 0 TYBOT(1) = (TYBOT(I)) +
    20 1 ((B(I,J)*H(I,J)) * YBAR(I,J))
    21 0 YBOT(I) = TYBOT(I)/AREA(I)
C 0000 0 CALCULATE DISTANCE FROM TOP
C 0000 0 TO CENTROID OF EACH SECTION
    22 0 DO 27 I=2,N
    23 0 DEPTH(I) = 0.0
    24 0 L = M(I)
    25 0 DO 26 J=1,L
    26 0 DEPTH(I) = DEPTH(I) + H(I,J)
    27 0 YTOP(I) = DEPTH(I) - YBOT(I)
C 0000 0 CALCULATE DISTANCE FROM
C 0000 1 CENTROID OF EACH PLATE TO THE
C 0000 2 CENTROID OF THE SECTION
C 0000 3 YPLATE
    28 0 DO 31 I =2,N
    29 0 L = M(I)
    30 0 DO 31 J=1,L
    31 0 YPLAT(I,J) = YBOT(I) -
    31 1 YBAR(I,J)
C 0000 0 CALCULATE MOMENT OF
C 0000 1 INERTIA -EYE
    32 0 DO 36 I=2,N
    33 0 EYE(I) = 0.0
    34 0 L = M(I)
    35 0 DO 36 J=1,L
    36 0 EYE(I) = EYE(I) + (B(I,J) *
    36 1 (H(I,J)**3.)) / 12. + B(I,J) *
    36 2 H(I,J) * (YPLAT(I,J)**2.)
C 0000 0 CALCULATE THE SECTION MODULI
C 0000 1 SXTOP AND SXBOT
    37 0 DO 41 I=2,N
    38 0 SXTOP(I) = EYE(I)/YTOP(I)
    39 0 SXBOT(I) = EYE(I)/YBOT(I)
C 0000 0 PUNCH RESULTS
    40 0 PUNCH, EYE(I), AREA(I)
    41 0 PUNCH, DEPTH(I), SXTOP(I),
    41 1 SXBOT(I)
    42 0 GO TO 2
    43 0 END
```

Figure 2-4

FORTRAN PROGRAM - Section Properties

CHAPTER III

ANGULAR FUNCTIONS

3-1 Discussion

Angular functions (flexibilities, carry-over flexibilities, and load functions) are end slopes due to unit cause or due to loads (Table 3-1). Numerous procedures are available for evaluating these functions:

1. Integral evaluation with unequal length segments.
2. Integral evaluation with equal length segments.
3. Conjugate beam method with unequal length segments.
4. Conjugate beam method with equal length segments.

It is the opinion of the author that an approach which yields not only the angular functions, but with additional steps will yield deflections, is desirable. One such method is that of the conjugate beam utilizing the string polygon (9). Advantages of using equal or unequal length strip segments are unique with each method. Equal length segments are perhaps easier to program and possess the advantage of allowing one to obtain deflections at equal spacings along the span. However, unequal length segments as determined by the cross sectional variation will yield a more accurate solution as compared with equal length segments, considering an equal number of segments to be used in both methods. When additional points of deflection

TERM	NAME	VALUE	VALUE FOR CONSTANT CROSS-SECTION	PHYSICAL MEANING	ILLUSTRATION
F_{ji}	ANGULAR FLEXIBILITY	$\int_j^i \frac{x^2 dx}{L_j^2 EI_x}$	$\frac{L_j}{3EI_j}$	end slope of a simple beam ij at j due to a unit moment applied at that end	
F_{jk}	ANGULAR FLEXIBILITY	$\int_j^k \frac{x'^2 dx}{L_k^2 EI_x}$	$\frac{L_k}{3EI_k}$	end slope of a simple beam jk at j due to a unit moment applied at that end	
G_{ij}	CARRY-OVER FLEXIBILITY	$\int_i^j \frac{xx' dx}{L_j^2 EI_x}$	$\frac{L_j}{6EI_j}$	end slope of a simple beam ij at i due to a unit moment applied at the far end j	
G_{kj}	CARRY-OVER FLEXIBILITY	$\int_k^j \frac{xx' dx}{L_k^2 EI_x}$	$\frac{L_k}{6EI_k}$	end slope of a simple beam jk at k due to a unit moment applied at the far end j	

τ_{ji}	ANGULAR LOAD FUNCTION	$\int_j^i \frac{BM_x x dx}{L_j EI_x}$		end slope of a simple beam ij at j due to loads	
τ_{jk}	ANGULAR LOAD FUNCTION	$\int_j^k \frac{BM_x x' dx}{L_k EI_x}$		end slope of a simple beam jk at j due to loads	

Table 3-1

Interpretation of Angular Functions

for the method of unequal segments are desired, auxiliary segments can be included. The author has thus chosen to use the conjugate beam approach utilizing the string polygon with unequal length segments.

3-2 Angular Flexibilities and Carry-Over Functions

3-2-1 Discussion

A typical span AB of a continuous beam of several spans is considered (Fig. 3-1). Angular flexibility F_{AB} and angular carry-over G_{BA} are obtained by applying a unit moment at A (Fig. 3-2). The reaction of the conjugate structure at A is F_{AB} and the reaction at B is G_{BA}. Likewise, by applying a unit moment at B on the real structure (Fig. 3-3), the reaction of the conjugate structure at B is angular flexibility F_{BA} and the reaction at A is angular carry-over G_{AB}. By virtue of Maxwell's Reciprocal Theorem, the angular carry-over at A is equal to the angular carry-over at B ($G_{AB} = G_{BA}$).

Span AB may be divided into n unequal segments. Applying the string polygon method (9), elastic weights for each segment are evaluated and applied as loads to the conjugate structure.

A typical elastic weight is expressed in terms of the three moment equation

$$\bar{P}_j = M_i G_{ij} + M_j \Sigma F_j + M_k G_{kj} + \Sigma \tau_j \tag{3-1}$$

The moment at various ordinates x due to either unit moment at A or B is a linear function of x.



Figure 3-1

Real Structure



Figure 3-2

Conjugate Structure - Unit Moment At A



Figure 3-3

Conjugate Structure - Unit Moment At B

$$M_i^A = \frac{x_i'}{L} \qquad M_i^B = \frac{x_i}{L}$$

$$M_j^A = \frac{x_j'}{L} \qquad M_j^B = \frac{x_j}{L} \tag{3-2}$$

$$M_k^A = \frac{x_k'}{L} \qquad M_k^B = \frac{x_k}{L}$$

and the end slopes due to loads are zero

$$\sum \tau_j^A = 0 \qquad \sum \tau_j^B = 0$$

Thus, from eq. 3-1

$$\bar{P}_j^A = \frac{x_i'}{L} G_{ij} + \frac{x_j'}{L} \Sigma F_j + \frac{x_k'}{L} G_{kj} \tag{3-3}$$

and

$$\bar{P}_j^B = \frac{x_i}{L} G_{ij} + \frac{x_j}{L} \Sigma F_j + \frac{x_k}{L} G_{kj} \tag{3-4}$$

where:

$\bar{P}_j^A$ - elastic weight at section j due to a unit moment at A

$\bar{P}_j^B$ - elastic weight at section j due to a unit moment at B

G_{ij} - angular carry-over flexibility of segment ij

G_{kj} - angular carry-over flexibility of segment kj

$\Sigma F_j = F_{ji} + F_{jk}$ - sum of the angular flexibilities of segments ji and jk

The conjugate reactions are then

$$F_{AB} = \sum_{j=1}^{n} \bar{P}_j^A \frac{x_j'}{L}$$

$$G_{BA} = \sum_{j=1}^{n} \bar{P}_j^A \frac{x_j}{L} \tag{3-5}$$

and

$$F_{BA} = \sum_{j=1}^{n} \bar{P}_j^B \frac{x_j}{L}$$

$$G_{AB} = \sum_{j=1}^{n} \bar{P}_j^B \frac{x_j'}{L} \tag{3-6}$$

3-2-2 Segment of Constant Cross Section

Considering typical segment ij (Fig. 3-4) of constant cross section, the following refinements may be made for angular flexibilities and carry-over functions (Eq's. 3-7).



Figure 3-4

Segment of Constant Cross Section

$$F_{ij} = F_{ji} = F_j = \frac{L_j}{3EI_j}$$

$$G_{ij} = G_{ji} = G_j = \frac{L_j}{6EI_j} \tag{3-7}$$

or,

$$F_j = \frac{L}{EI_o} \cdot \frac{L_j I_o}{3LI_j} = \frac{L}{EI_o} F_j'$$

$$G_j = \frac{L}{EI_o} \cdot \frac{L_j I_o}{6LI_j} = \frac{L}{EI_o} G_j' \tag{3-8}$$

where:

$$F_j' = \frac{L_j I_o}{3LI_j}$$

$$G_j' = \frac{L_j I_o}{6LI_j}$$

This assumption is reasonably justified for sections whose cross section is not constant, provided average dimensions are taken for the plate dimensions.

From Eq's. 3-3, 4, the expression for a typical elastic weight due to unit moment at A or B respectively, is

$$\bar{P}_j^A = \frac{L}{EI_o}\left[\frac{x_i'}{L} G_j' + \frac{x_j'}{L}(F_j' + F_k') + \frac{x_k'}{L} G_k'\right] \tag{3-9}$$

$$\bar{P}_j^B = \frac{L}{EI_o}\left[\frac{x_i}{L} G_j' + \frac{x_j}{L}(F_j' + F_k') + \frac{x_k}{L} G_k'\right]$$

From Eq's. 3-5, 6 , the end reactions of the conjugate beam become

$$F_{AB} = \frac{L}{EI_o} F_{AB}'$$

$$F_{BA} = \frac{L}{EI_o} F_{BA}' \tag{3-10}$$

$$G_{AB} = G_{BA} = \frac{L}{EI_o} \; G_{AB}'$$

3-3 Angular Load Functions

3-3-1 Discussion

By definition, angular load functions are end slopes due to loads (Table 3-1). Mathematically stated:

$$\tau_{AB} = \int_A^B \frac{BM_x \, x' \, dx}{LEI_x}$$

$$\tau_{BA} = \int_A^B \frac{BM_x \, x \, dx}{LEI_x}$$

where:

BM_x = the bending moment at section x of the simple span AB due to loads.

Those cases of loading considered in this thesis include:

1. Unit load applied at each cut-off ordinate (live load)
2. Uniform load
3. Dead load

3-3-2 Unit Load

The angular load functions due to a unit load at each cut-off ordinate is useful not only for evaluating influence line ordinates for live loads, but also for evaluating the angular load functions due to uniform load and dead load.

The moment at any ordinate along span AB due to a unit moment at A (or B) is a linear function of that ordinate (Eq. 3-2).

The conjugate shear at A, $\overline{V}_{AB}$, (Fig. 3-6) due to a unit moment at A or B, respectively, is

$$\overline{V}_{AB}^{\,A} = \sum_{j=1}^{n} \overline{P}_j^A \frac{x_j'}{L} = \frac{L}{EI_o} F_{AB}'$$

and (3-11)

$$\overline{V}_{AB}^{\,B} = \sum_{j=1}^{n} \overline{P}_j^B \frac{x_j'}{L} = \frac{L}{EI_o} G_{AB}'$$

The conjugate shear at a typical ordinate, $\overline{V}_j$, (Fig. 3-6) due to a unit moment at A or B, respectively, is

$$\overline{V}_j^A = \overline{V}_i^A - \overline{P}_i^A = \frac{L}{EI_o} \overline{V}_j^{A\prime}$$

and (3-12)

$$\overline{V}_j^B = \overline{V}_i^B - \overline{P}_i^B = \frac{L}{EI_o} \overline{V}_j^{B\prime}$$

which by virtue of the theory of the conjugate structure is equal to the slope of the corresponding section of the real structure.

$$\overline{V}_j^A = \Theta_j^A$$



(a)

real structure



(b)

moment diagram due to $M_B = 1$

(c)

moment diagram due to $M_A = 1$

Figure 3-5

Real Structure With Moment Diagrams Due To Unit End Moments



(a)

conjugate structure



(b)

conjugate shear diagram



(c)

conjugate moment diagram

Figure 3-6

Conjugate Structure With Shear And Moment Diagrams

$$\bar{V}_j^B = \Theta_j^B$$

The conjugate bending moment at any ordinate, $\bar{M}_j$, (Fig. 3-6) due to a unit moment at A or B, respectively, can then be expressed

$$\bar{M}_j^A = \bar{M}_i^A + \bar{V}_j^A L_j$$

$$= \frac{L^2}{EI_o} \bar{M}_j^{A\prime}$$

and (3-13)

$$\bar{M}_j^B = \bar{M}_i^B + \bar{V}_j^B L_j$$

$$= \frac{L^2}{EI_o} \bar{M}_j^{B\prime}$$

which by virtue of the theory of the conjugate structure is equal to the deflection of the corresponding section of the real structure. Thus,

$$\bar{M}_j^A = \Delta_j^A$$

$$\bar{M}_j^B = \Delta_j^B$$

From Maxwell's Reciprocal Principle, the deflection at section j due to a unit moment at A ($M_A = 1$) is numerically equal to the end slope at A due to a unit force at section j. Thus,

$$\bar{M}_j^A = \Delta_j^A = (\tau_{AB})_j$$

and (3-14)

$$\bar{M}_j^B = \Delta_j^B = (\tau_{BA})_j$$

where: $(\tau_{AB})_j$ - angular load function (end slope) at A due to a unit force applied at section j

$(\tau_{BA})_j$ - angular load function (end slope) at B due to a unit force applied at section j

3-3-3 Uniform Load

Typical span AB is loaded by a uniform load of intensity w (Fig. 3-7).



Figure 3-7

Uniform Loading

For a typical concentrated load P_j^{UL}, one half the load of segments j and k are considered to be concentrated at j.

$$P_j^{UL} = w\left[\frac{L_j}{2} + \frac{L_k}{2}\right]$$

The angular load function for uniform load is obtained by multiplying the equivalent concentrated loads by the corresponding end slope influence line ordinates (Eq's. 3-15).

$$\tau_{AB}^{UL} = \sum_{j=1}^{n} P_j^{UL} \cdot (\tau_{AB})_j$$

$$\tau^{UL}_{AB} = \frac{wl^3}{EI_o} \tau^{UL'}_{AB}$$

$$\tau^{UL}_{BA} = \sum_{j=1}^{n} P^{UL}_{j} \cdot (\tau_{BA})_j \qquad (3\text{-}15)$$

$$\tau^{UL}_{BA} = \frac{wl^3}{EI_o} \tau^{UL'}_{BA}$$

where: w - uniform load intensity per unit length

τ^{UL}_{AB} - angular load function (end slope) at A due to uniform load applied to span AB

τ^{UL}_{BA} - angular load function (end slope) at B due to uniform load applied to span AB

3-3-4 Dead Load

Typical span AB is loaded by dead load (Fig. 3-8). If typical segment ij is considered to be of uniform cross section (average plate dimensions used for calculating cross sectional area), it may be considered to be loaded by a uniform load of magnitude w_j.



Figure 3-8

Dead Load

For a typical concentrated load P_j, one half the load of segments j and k are considered to be concentrated at j.

$$P_j^{DL} = \rho A_o \left[\frac{A_j}{A_o} \frac{L_j}{2} + \frac{A_k}{A_o} \frac{L_k}{2} \right]$$

The angular load function for dead load is obtained by multiplying the equivalent concentrated loads by the corresponding end slope influence line ordinates (Eq's. 3-16).

$$\tau_{AB}^{DL} = \sum_{j=1}^{n} P_j^{DL} \cdot (\tau_{AB})_j$$

$$= \frac{\rho A_o L^3}{EI_o} \tau_{AB}^{DL'}$$

$$\tau_{BA}^{DL} = \sum_{j=1}^{n} P_j^{DL} \cdot (\tau_{BA})_j \qquad (3\text{-}16)$$

$$= \frac{\rho A_o L^3}{EI_o} \tau_{BA}^{DL'}$$

where:
- ρ - unit weight of construction material
- τ_{AB}^{DL} - angular load function (end slope) at A due to dead load of span AB
- τ_{BA}^{DL} - angular load function (end slope) at B due to dead load of span AB
- A_o - reference area

3-4 Computer Program

3-4-1 Results of Program

The Phase II computer program (Fig's. 3-9, 10) is written for the IBM 650 electronic computer using FORTRAN language and floating point arithmethic The program yields a solution for:

FABPM - angular flexibility coefficient at end A of span AB

FBAPM - angular flexibility coefficient at end B of span AB

GAORB - angular carry-over coefficient at either end A or B of span AB

BMA(I) - conjugate moment coefficient at section i of the conjugate structure due to unit moment at A(M_A = 1); also the deflection coefficient of the real structure at i due to unit moment at A; also the angular load function coefficient at A due to a unit force applied at section i

BMB(I) - conjugate moment coefficient at section i of the conjugate structure due to unit moment at B (M_B = 1); also the deflection coefficient of the real structure at i due to unit moment at B; also the angular load function coefficient at A due to a unit force applied at section i

TABUL - angular load function coefficient at A due to uniform loading

TBAUL - angular load function coefficient at B due to uniform loading

TABDL - angular load function coefficient at A due to dead load

TBADL - angular load function coefficient at B due to dead load

3-4-2 Data Required

The input data necessary for computing the desired angular functions include:

N - total number of segments

S(I) - distance from left end of span AB to the cut-off point at i divided by the span length L $(\frac{x_i}{L})$

EYE(I) - moment of inertia of segment i

AREA(I) - cross sectional area of segment i

The quantities EYE(I) and AREA(I) are read directly from the results of the section properties program (Phase I).

3-4-3 Auxiliary Quantities

Other quantities integral in the program are denoted:

FPRIM(I) - flexibility coefficient of segment i

GPRIM(I) - angular carry-over coefficient of segment i

PRIMA(I) - individual elastic weight coefficient at i due to a unit moment at A ($M_A = 1$)

PRIMB(I) - individual elastic weight coefficient at i due to a unit moment at B ($M_B = 1$)

VBARA(I) - conjugate shear coefficient at section i of the conjugate structure due to unit moment at A ($M_A = 1$)

VBARB(I) - conjugate shear coefficient at section i of the conjugate structure due to unit moment at B ($M_B = 1$)

EYEO - reference moment of inertia (taken as EYE(2) in computer program)

AREAO - reference area (taken as AREA(2) in computer program)



Figure 3-9

Flow Chart Phase II - Angular Functions

```
C 0000 0 PHASE 2
C 0000 0 EVALUATION OF ANGULAR FUNCTION
C 0000 0 ANGULAR FLEXIBILITIES AND
C 0000 0 CARRY OVER VALUES
    1 0 DIMENSION S(25), EYE(25),
    1 1 FPRIM(25), GPRIM(25),
    1 2 PRIMA(25), PRIMB(25), AREA(25)
    1 3 , VBARA(25), VBARB(25)
    1 4 ,BMA(25), BMB(25)
    2 0 READ, N
    3 0 READ, (S(I), I=1,N)
    4 0 DO 5 I=2,N
    5 0 READ, EYE(I), AREA(I)
    6 0 EYEO = EYE(2)
        AREAO = AREA(2)
    7 0 DO 9 I=2,N
    8 0 FPRIM(I) = ((S(I) - S(I-1)) *
    8 1 EYEO) / (3. * EYE(I))
    9 0 GPRIM(I) = FPRIM(I) / 2.
   10 0 K = N-1
   11 0 DO 13 I=2,K
   12 0 PRIMB(I) = S(I-1) * GPRIM(I)
   12 1 + S(I) * (FPRIM(I) + FPRIM(I+1
   12 2 )) + S(I+1) * GPRIM(I+1)
   13 0 PRIMA(I) = (1.-S(I-1)) *GPRIM
   13 1 (I) + (1.-S(I)) * (FPRIM(I) +
   13 2 FPRIM(I+1)) +(1. - S(I+1)) *
   13 3 GPRIM(I+1)
   14 0 PRIMB(1) = S(2) * GPRIM(2)
   15 0 PRIMA(1) = FPRIM(2) + (1.-S(2)
   15 1 ) * GPRIM(2)
   16 0 PRIMB(N) = S(N-1) * GPRIM(N)
   16 1 + FPRIM(N)
   17 0 PRIMA(N) = (1.-S(N-1)) *
   17 1 GPRIM(N)
   18 0 FABPM = 0.0
   19 0 FBAPM = 0.0
   20 0 GAORB = 0.0
   21 0 DO 24 I=1,N
   22 0 FABPM = FABPM + (PRIMA(I) *
   22 1 (1.-S(I)))
   23 0 FBAPM = FBAPM + (PRIMB(I) *
   23 1 S(I))
   24 0 GAORB = GAORB + (PRIMB(I) *
   24 1 (1.-S(I)))
   25 0 PUNCH, FABPM, FBAPM, GAORB
C 0000 0 CASE OF UNIT LOAD
C 0000 0 CALCULATE MODIFIED ANGULAR
C 0000 0 LOAD FUNCTIONS DUE TO UNIT
C 0000 0 FORCE APPLIED AT SECTION I.
C 0000 0 ALSO MODIFIED DEFLECTION OF
C 0000 0 SECTION I DUE TO UNIT END
C 0000 0 MOMENT
   26 0 VBARA(2) = FABPM - PRIMA(1)
   27 0 VBARB(2) = GAORB - PRIMB(1)
   28 0 DO 30 I=3,N
   29 0 VBARA(I) = VBARA(I-1) -
   29 1 PRIMA(I-1)
   30 0 VBARB(I) = VBARB(I-1)
   30 1 - PRIMB(I-1)
   31 0 BMA(1) = 0.0
   32 0 BMB(1) = 0.0
   33 0 DO 35 I=2,N
   34 0 BMA(I) = BMA(I-1) + (S(I) -
   34 1 S(I-1)) * VBARA(I)
   35 0 BMB(I) = BMB(I-1) + (S(I)
   35 1 - S(I-1)) * VBARB(I)
   36 0 PUNCH, (BMA(I), I=1,N)
   37 0 PUNCH, (BMB(I), I=1,N)
C 0000 0 CASE OF UNIFORM LOAD
C 0000 0 CALCULATE MODIFIED ANGULAR
C 0000 0 FUNCTIONS DUE TO UNIFORM LOAD
   38 0 TABUL = 0.0
   39 0 TBAUL = 0.0
   40 0 DO 42 I=2,K
   41 0 TABUL = TABUL + ((S(I+1) -
   41 1 S(I-1)) / 2.0) * BMA(I)
   42 0 TBAUL = TBAUL + ((S(I+1) -
   42 1 S(I-1)) / 2.) * BMB(I)
   43 0 PUNCH, TABUL, TBAUL
C 0000 0 CASE OF DEAD LOAD
C 0000 0 CALCULATE MODIFIED ANDULAR
C 0000 0 FUNCTIONS DUE TO DEAD LOAD
   44 0 TABDL = 0.0
   45 0 TBADL = 0.0
   46 0 DO 48 I=2,K
   47 0 TABDL = TABDL+(AREA(I)/AREAO *
   47 1 (S(I)-S(I-1))/2.0 + AREA(I+1)
   47 2 /AREAO * (S(I+1)-S(I)) / 2.0)*
   47 3 BMA(I)
   48 0 TBADL = TBADL + (AREA(I)/AREAO
   48 1 * (S(I)-S(I-1)) / 2.0 + AREA(
   48 2 I+1)/ AREAO * (S(I+1)-S(I)) /
   48 3 2.0) * BMB(I)
   49 0 PUNCH, TABDL, TBADL
   50 0 DO 51 I=2,N
   51 0 PUNCH, FPRIM(I), GPRIM(I)
   52 0 GO TO 2
   53 0 END
```

Figure 3-10

FORTRAN PROGRAM - Angular Functions

CHAPTER IV

DEFLECTIONS

4-1 Discussion

The total deflection Δ_x at any section x due to loads can be expressed

$$\Delta_x = B\Delta_x + M_A(\tau_{AB})_x + M_B(\tau_{BA})_x \qquad (4\text{-}1)$$

where: Δ_x - total deflection at ordinate x due to loads

$B\Delta_x$ - deflection at ordinate x of simple beam AB due to loads

M_A - bending moment at support A due to loads

M_B - bending moment at support B due to loads

$(\tau_{AB})_x$ - deflection at ordinate x of simple beam AB due to $M_A = 1$

$(\tau_{BA})_x$ - deflection at ordinate x of simple beam AB due to $M_B = 1$

A numerical solution for deflection can be obtained only after Phase II has been completed for angular functions. A solution for the final end moments can be obtained by substituting into the three moment equation. The simple beam deflections due to unit moment at A or B are BMA(I) and BMB(I) respectively, from the computer program for angular functions.

The simple beam deflection due to loads is developed as Phase

III. Two types of loading are considered:

1. uniform load
2. dead load

4-2 Uniform Load

Typical span AB is loaded by a uniform load (Fig. 4-1), causing deflection of the real structure.

By virtue of the theory of the conjugate beam, the simple beam deflections due to uniform load can be evaluated as the magnitude of the conjugate moment diagram.

A typical elastic weight is from Equation 3-1

$$\bar{P}_j^{UL} = M_i^{UL} \; G_{ij} + M_j^{UL} \sum F_j + M_k^{UL} G_{kj} + \sum \tau_j$$

The bending moment diagram for a simple beam with uniform loading is represented by a second degree parabola. A typical moment may be expressed as

$$M_j^{UL} = \frac{wL^2}{8} \left[4 \frac{x_j}{L} \left(1 - \frac{x_j}{L}\right) \right]$$

Thus a typical elastic weight is

$$\bar{P}_j^{UL} = \frac{wL^2}{8} \left[4 \frac{x_i}{L} \left(1 - \frac{x_i}{L}\right) G_j + 4 \frac{x_j}{L} (F_j + F_k)\left(1 - \frac{x_j}{L}\right) \right.$$

$$\left. + 4 \frac{x_k}{L}\left(1 - \frac{x_k}{L}\right) G_k \right] + \left[\frac{wL_j^3}{24EI_j} + \frac{wL_k^3}{24EI_k} \right]$$

$$\bar{P}_j^{UL} = \frac{wL^3}{24EI_o} \; \bar{P}_j^{UL'} \qquad (4\text{-}2)$$

The conjugate end shear (Fig. 4-2) is



Figure 4-1

Deflection Due To Uniform Load



Figure 4-2

Conjugate Structure Loading With Shear And Moment Diagrams

$$\bar{V}_{AB}^{UL} = \sum_{j=1}^{n} \bar{P}_j^{UL} \frac{x_j'}{L}$$

$$\bar{V}_{AB}^{UL} = \frac{wL^3}{24EI_o} \sum_{j=1}^{n} \frac{x_j'}{L} \bar{P}_j^{UL'}$$

$$\bar{V}_{AB}^{UL} = \frac{wL^3}{24EI_o} \bar{V}_{AB}^{UL'} \tag{4-3}$$

and the conjugate shear at any section is

$$\bar{V}_j^{UL} = \bar{V}_i^{UL} - \bar{P}_i^{UL}$$

$$\bar{V}_j^{UL} = \frac{wL^3}{24EI_o} \bar{V}_j^{UL'} \tag{4-4}$$

The conjugate moment at any section is then the simple beam deflection due to uniform load.

$$B\Delta_j^{UL} = \bar{M}_j^{UL} = \bar{M}_i^{UL} + \bar{V}_j^{UL} L_j$$

$$B\Delta_j^{UL} = \frac{wL^4}{24EI_o} B\Delta_j^{UL'} \tag{4-5}$$

<u>4-3 Dead Load</u>

Typical span AB is loaded by dead load (Fig. 4-3), causing deflection of the real structure. If typical segment ij is considered to be of uniform cross section as in Chapters 2 and 3, the segment is loaded by uniform load of magnitude w_j where

$$w_j = \rho A_j$$

To calculate end shear V_{AB}^{DL}, the total uniform load of typical segment ij is considered concentrated at the center of the segment. End shear V_{AB}^{DL} can then be evaluated.

$$V_{AB}^{DL} = \sum_{j=1}^{n} P_j^{DL} \left(\frac{x_j'}{L} + \frac{L_j}{2L}\right)$$

$$V_{AB}^{DL} = L \rho A_o \sum_{j=1}^{n} \left[\frac{A_j}{A_o} \cdot \frac{L_j}{L} \left(\frac{x_j'}{L} + \frac{L_j}{2L}\right)\right]$$

$$V_{AB}^{DL} = L \rho A_o V_{AB}^{DL'} \qquad (4\text{-}6)$$

where: $P_j^{DL} = w_j L_j = \rho A_j \cdot L_j$ - total weight of segment ij

ρ - unit weight of construction material

A_j - cross sectional area of segment ij

A_o - reference area

The moment at any ordinate j is equal in magnitude to the area of the shear diagram to that point. Although segment ij is loaded by uniform load w_j, the shear at ordinate j is equal to the shear at section i minus the total load on section ij, P_j^{DL}.

$$V_j^{DL} = V_i^{DL} - P_j^{DL}$$

$$V_j^{DL} = L \rho A_o V_{AB}^{DL'} \qquad (4\text{-}7)$$

And the moment at any ordinate j is:

$$M_j^{DL} = M_i + \frac{L_j}{2} (V_i^{DL} + V_j^{DL})$$



Figure 4-3

Deflection Due To Dead Load



Figure 4-4

Conjugate Structure Loading With Shear And Moment Diagrams

$$\bar{M}_j^{DL} = L^2 \rho A_o \bar{M}_j^{DL'} \tag{4-8}$$

A typical elastic weight due to dead load (Fig. 4-4) is then from Equation 3-1

$$\bar{P}_j^{DL} = \frac{L}{EI_o} \left[M_i^{DL} G_i' + M_j^{DL} (F_j' + F_k') + M_k^{DL} G_k' \right] + \Sigma \tau_j^{DL}$$

$$\bar{P}_j^{DL} = \frac{\rho A_o L^3}{24EI_o} \bar{P}_j^{DL'} \tag{4-9}$$

where:

$$\Sigma \tau_j^{DL} = \tau_j^{DL} + \tau_k^{DL}$$

$$= \frac{A_j \rho L_j^3}{24EI_j} + \frac{A_k \rho L_k^3}{24EI_k}$$

$$= \frac{\rho A_o L^3}{24EI_o} \left[\frac{A_j}{A_o} \left(\frac{L_j}{L}\right)^3 \frac{I_o}{I_j} + \frac{A_k}{A_o} \left(\frac{L_k}{L}\right)^3 \frac{I_o}{I_k} \right]$$

The conjugate end shear (Fig. 4-4) is

$$\bar{V}_{AB}^{DL} = \sum_{j=1}^{n} \bar{P}_j^{DL} \frac{x_j'}{L}$$

$$\bar{V}_{AB}^{DL} = \frac{\rho A_o L^3}{24EI_o} \sum_{j=1}^{n} \bar{P}_j^{DL'} \frac{x_j'}{L}$$

$$\bar{V}_{AB}^{DL} = \frac{\rho A_o L^3}{24EI_o} \bar{V}_{AB}^{DL'} \tag{4-10}$$

and the conjugate shear at any section is

$$\bar{V}_j^{DL} = \bar{V}_i^{DL} - \bar{P}_j^{DL}$$

$$\bar{V}_j^{DL} = \frac{\rho A_o L^3}{24EI_o} \bar{V}_j^{DL'} \qquad (4\text{-}11)$$

The conjugate moment at any section is then the simple beam deflection due to uniform load.

$$B\Delta_j^{DL} = \bar{M}_j^{DL} = \bar{M}_i^{DL} - \bar{V}_j^{DL} L_j$$

$$B\Delta_j^{DL} = \frac{\rho A_o L^4}{24EI_o} B\Delta_j^{DL'} \qquad (4\text{-}12)$$

4-4 Computer Program

4-4-1 Results Of Program

The program presented (Fig's. 4-5, 6) is written for the IBM 650 electronic computer using FORTRAN language and floating point arithmetic. The program yields a solution for:

BMUL(I) - deflection coefficient at ordinate i of simple span AB due to uniform loading

BMDL(I) - deflection coefficient at ordinate i of simple span AB due to dead loading

4-4-2 Data Required

The input data necessary for computing these deflections are

N - total number of segments

S(I) - distance from left end of member AB to the cut-off point at i divided by the span length L - $\frac{x_i}{L}$

EYE(I) - moment of inertia of segment i

AREA(I) - cross sectional area of segment i

FPRIM(I) - angular flexibility coefficient of segment i

GPRIM(I) - angular carry-over flexibility coefficient of segment i

The quantities EYE(I) and AREA(I) are answers from Phase I (Sectional Properties). The quantities FPRIM(I) and GPRIM(I) are answers from Phase II (Angular Functions). The program has been written so as to take these answer cards directly as data for Phase III (Deflections).

4-4-3 Auxiliary Quantities

Other quantities integral in the program include:

PBRUL(I) - elastic weight coefficient due to uniform loading

VABPM - end shear coefficient at A due to dead load

VPM(I) - shear coefficient at section i due to dead load

BMPM(I) - moment coefficient at section i due to dead load

PBRDL(I) - elastic weight coefficient at section i due to dead load

VABDL(I) - conjugate shear coefficient at A due to dead load

VABUL(I) - conjugate shear coefficient at A due to uniform load

VDL(I) - conjugate shear coefficient at section i due to dead load

VUL(I) - conjugate shear coefficient at section i due to uniform load

EYEO - reference moment of inertia (taken as EYE(2) in computer program)

AREAO - reference area (taken as AREA(2) in computer program)



Figure 4-5

Flow Chart Phase III- Deflections

```
C 0000 0 PHASE 3
C 0000 0 SIMPLE BEAM DEFLECTION
C 0000 0 FOR DEAD LOAD AND UNIFORM LOAD
C 0000 0 ELASTIC WEIGHT DUE TO UNIFORM
C 0000 0 LOAD
     1 0 DIMENSION S(25), EYE(25),
     1 1 FPRIM(25), GPRIM(25),
     1 2 AREA(25), PBRUL(25)
     2 0 DIMENSION VPM(25), BMPM(25),
     2 1 PBRDL(25), VDL(25), VUL(25),
     2 2 BMDL(25), BMUL(25)
     3 0 READ, N
     4 0 READ, (S(I), I=1,N)
     5 0 DO 6 I=2,N
     6 0 READ, EYE(I), AREA(I)
     7 0 DO 8 I=2,N
     8 0 READ, FPRIM(I), GPRIM(1)
         EYEO = EYE(2)
         AREAO = AREA(2)
         K = N-1
     9 0 DO 11 I=2,K
    10 0 PBRUL(I) = (4./3.) * (S(I-1) *
    10 1 (1.-S(I-1)) * GPRIM(I) + S(I)
    10 2 * (1.-S(I)) * (FPRIM(I) +
    10 3 FPRIM(I+1)) + S(I+1) *
    10 4 (1.-S(I+1)) * GPRIM(I+1))
    11 0 PBRUL(I) = PBRUL(I) + (EYEO/
    11 1 EYE(I)) * ((S(I) - S(I-1))**3.
    11 2 ) + (EYEO/EYE(I+1)) * ((
    11 3 S(I+1)-S(I))**3.)
    12 0 PBRUL(1) = (4./3.) * S(2) *
    12 1 (1.-S(2)) * GPRIM(2) + (EYEO/
    12 2 EYE(2)) * (S(2)**3.)
    13 0 PBRUL(N) = (4./3.) * S(N-1) *
    13 1 (1.-S(N-1)) * GPRIM(N)+ (EYEO/
    13 2 EYE(N)) * ((S(N) - S(N-1))
    13 3 **3.)
C 0000 0 ELASTIC WEIGHT DUE TO DEAD
C 0000 0 LOAD
    14 0 VABPM = 0.0
    15 0 DO 16 I=2,N
    16 0 VABPM = VABPM + AREA(I)/AREAO*
    16 1 (S(I)-S(I-1)) * ((1.-S(I)) +
    16 2 (.5* (S(I)-S(I-1))))
    17 0 VPM(1) = VABPM
    18 0 DO 19 I=2,N
    19 0 VPM(I) = VPM(I-1) - AREA(I) /
    19 1 AREAO * (S(I)-S(I-1))
    20 0 BMPM(1) = 0.0
    21 0 DO 22 I=2,N
    22 0 BMPM(I) = BMPM(I-1) + .5 *
    22 1 (S(I)-S(I-1)) * (VPM(I-1) +
    22 2 VPM(I))
    23 0 DO 25 I=2,K
    24 0 PBRDL(I) = AREA(I)/AREAO *((S
    24 1 (I) -S(I-1))**3.) * (EYEO/EYE
    24 2 (I)) +AREA(I+1)/AREAO * ((S(I+
    24 3 1)-S(I))**3.)*(EYEO/EYE(I+1))
    25 0 PBRDL(I) = PBRDL(I) + 24.*
    25 1 (BMPM(I-1) * GPRIM(I-1) +
    25 2 BMPM(I) * (FPRIM(I) +
    25 3 FPRIM(I+1)) + BMPM(I+1)*GPRIM(
    25 4 I+1))
    26 0 PBRDL(1) = 24.* BMPM(2) *
    26 1 GPRIM(2)+AREA(2)/AREAO * (S(2)
    26 2 **3.) * (EYEO/EYE(2))
    27 0 PBRDL(N) = 24.* BMPM(N-1) *
    27 1 GPRIM(N-1) + AREA(N) / AREAO *
    27 2 ((S(N)-S(N-1))**3.) * (EYEO /
    27 3 EYE(N))
    28 0 VABDL = 0.0
    29 0 VABUL = 0.0
    30 0 DO 32 I=1,N
    31 0 VABDL = VABDL+(1.-S(I)) *
    31 1 PBRDL(I)
    32 0 VABUL = VABUL + (1.-S(I)) *
    32 1 PBRUL(I)
    33 0 VDL(1) = VABDL
    34 0 VUL(1) = VABUL
    35 0 DO 37 I=2,N
    36 0 VDL(I) = VDL(I-1) - PBRDL(I)
    37 0 VUL(I) = VUL(I-1) - PBRUL(I)
    38 0 BMDL(1) = 0.0
    39 0 BMUL(1) = 0.0
    40 0 DO 42 I=2,N
    41 0 BMDL(I) = BMDL(I-1) + (S(I) -
    41 1 S(I-1)) * VDL(I)
    42 0 BMUL(I) = BMUL(I-1) + (S(I) -
    42 1 S(I-1)) * VUL(I)
    43 0 PUNCH, (BMDL(I), I=2,N)
    44 0 PUNCH, (BMUL(I), I=2,N)
    45 0 GO TO 3
    46 0 END
```

Figure 4-6

FORTRAN PROGRAM - Deflections

CHAPTER V

APPLICATION

5-1 Introduction

One span from the five span continuous beam Benton Street Bridge in Iowa City, Iowa (1,5) is considered (Fig. 5-1, 2). The bridge, designed by Ned L. Ashton, is an all welded steel deck girder highway bridge completed in July, 1949.

Sectional properties, angular function coefficients, and deflection coefficients are computed using the FORTRAN programs from Chapters 2, 3, 4. Available known results check favorably. Coefficients must of course be multiplied by the corresponding constants to obtain final angular functions or deflections.

5-2 Cross Sectional Properties

Data is presented in inches or dimension-less quantities and are of the form:

$$\begin{array}{l} N \\ M(I) \\ \left. B(I,J),\ H(I,J),\quad J = 1,\ M(I) \right] \ I = 2,8 \end{array}$$



Figure 5-1
Application Problem



Figure 5-2
Cross Sectional Dimensions

```
          8+
          3+
1800000052+  1250000051+
5000000050+  4550000052+
1800000052+  1250000051+
          5+
1700000052+  3593750050+
1800000052+  1250000051+
5000000050+  4550000052+
1800000052+  1250000051+
1700000052+  3593750050+
          5+
1700000052+  5000000050+
1800000052+  1250000051+
5000000050+  4550000052+
1800000052+  1250000051+
1700000052+  5000000050+
          5+
1700000052+  5000000050+
1800000052+  1250000051+
5000000050+  4550000052+
1800000052+  1250000051+
1700000052+  5000000050+
          3+
1800000052+  1250000051+
5000000050+  4550000052+
1800000052+  1250000051+
          3+
1800000052+  2000000051+
5000000050+  4400000052+
1800000052+  2000000051+
          5+
1700000052+  7500000050+
1800000052+  2000000051+
5000000050+  4400000052+
1800000052+  2000000051+
1700000052+  7500000050+
```

Computed answers are of the form:

$$\left.\begin{array}{l}\text{EYE(I), AREA(I)}\\ \text{DEPTH(I), SXTOP(I), SXBOT(I)}\end{array}\right] \text{I = 2, 8}$$

```
2851832055+  6775000052+
4800000052+  1188263354+  1188263354+
3566224355+  7996875052+
4871875052+  1464004854+  1464004954+
3851574855+  8475000052+
4900000052+  1572071354+  1572071354+
3851574855+  8475000052+
4900000052+  1572071354+  1572071354+
2851832055+  6775000052+
4800000052+  1188263354+  1188263354+
4166136955+  9400000052+
4800000052+  1735890454+  1735890454+
5681317755+  1195000053+
4950000052+  2295482054+  2295481854+
```

5-3 Angular Functions

Sectional properties, moment of inertia and area, are read as data directly from Phase I (Cross-Sectional Properties). The data required is of the form:

N

S(I), I = 2,8

EYE(I), AREA(I), I = 2,8

```
          8+
             2051282050+ 2307692350+ 4358974350+ 6410256450+ 8205128250+
9166666650+ 1000000051+
2851832055+ 6775000052+
3566224355+ 7996875052+
3851574855+ 8475000052+
3851574855+ 8475000052+
2851832055+ 6775000052+
4166136955+ 9400000052+
5681317755+ 1195000053+
```

Computed answers are of the form:

FAB, FBA, GAORB

BMA(I), I = 2,8

BMB(I), I = 2,8

TABUL, TBAUL

TABDL, TBADL

FPRIM(I), GPRIM(I), I = 2,8

```
2941686450+ 2503127450+ 1360281550+
            4074204449+ 4335788049+ 5173109449+ 4252934349+ 2383419349+
1128728049+ 2100000043+
            3906222847+ 4333126947+ 6991551447+ 7645451547+ 5399194547+
2828764747+ 2804428041+
3279186749+ 3287989649+
2633633251+ 2654712951+
6837606649+ 3418803349+
6834857948+ 3417429048+
5062787849+ 2531393949+
5062788149+ 2531394149+
5982906049+ 2991453049+
2193995749+ 1096997949+
1394351249+ 6971756048+
```

5-4 Deflections

Answer cards FPRIM(I) and GPRIM(I) from Phase II (Angular Functions) are added to the data cards for Phase II to form data cards for Phase III (Deflections).

N

S(I), I = 2,8

EYE(I), AREA(I), I = 2,8

FPRIM(I), GPRIM(I), I = 2,8

```
        8+
            2051282050+ 2307692350+ 4358974350+ 6410256450+ 8205128250+
9166666650+ 1000000051+
2851832055+ 6775000052+
3566224355+ 7996875052+
3851574855+ 8475000052+
3851574855+ 8475000052+
2851832055+ 6775000052+
4166136955+ 9400000052+
5681317755+ 1195000053+
6837606649+ 3418803349+
6834857948+ 3417429048+
5062787849+ 2531393949+
5062788149+ 2531394149+
5982906049+ 2991453049+
2193995749+ 1096997949+
1394351249+ 6971756048+
```

Computed answers are of the form:

BMDL(I), I = 2,8

BMUL(I), I = 2,8

1859873350+ 2043632950+ 2857046250+ 2718139550+ 1649428250+ 8127633949+
8560885643+
2145895749+ 2347967349+ 3368698449+ 3164053249+ 1903821649+ 9138638048+
1330000043+

CHAPTER VI

SUMMARY AND CONCLUSION

6-1 Summary

In this thesis three separate programs for the IBM 650 electronic computer have been developed in FORTRAN language utilizing floating point arithmetic. The first program gives a solution for the sectional properties of a typical span from a continuous variable cross section structure. An upper limit of twenty five segments with nine plates per segment has been taken. The IBM 650 could handle a greater number of segments or plates if needed and correctly dimensioned. The second program gives the solution for angular functions for this span. The third program gives a solution for the deflections of this span due to dead and uniform loading.

6-2 Conclusion

The analysis of a continuous structure of variable cross section for cross sectional properties, angular functions, and deflections proves to be quite adaptable to solution by electronic computer. Once FORTRAN statements have been processed and an object program obtained, very little machine time is required for a typical solution.

The derivation and application is best suited for a plate girder type structure where the cross section is uniform between cut-off points However, when the cross section between cut-off points is not uniform, basic assumptions can be made for sectional dimensions, yielding very good results.

An unfortunate characteristic of obtaining deflections is that only those deflections where elastic weights are applied may be calculated. However, when additional considerations of deflection are desirable, additional elastic weights may be considered at these locations.

It is felt by the author that this work can be extended in subsequent research to include the evaluation of influence line ordinates and the development of a design procedure using the electronic computer. It is expected that this research would open the way for the development of an optimum design approach.

A SELECTED BIBLIOGRAPHY

1. Ashton, N. L., "A Modern Steel Deck Girder Highway Bridge", Studies in Structural Arc Welding, Lincoln Electric Company, Cleveland, Ohio, Bulletin 838, 1952.

2. Chu, Shin-Lung, "Beam Constants by the String Polygon Method", M. S. Thesis, Oklahoma State University, Stillwater, 1959.

3. Exline, J. W., "String Polygon Constants for Members With Sudden Change in Section", M. S. Thesis, Oklahoma State University, Stillwater, 1961.

4. Gillespie, J. W., Gauger, F. N., and Tuma, J. J., "Volume IV, Dead Load Deflections by High Speed Computer", School of Civil Engineering Research Publication, Oklahoma State University, Stillwater, No. 6, 1960.

5. Havner, K. S., and Tuma, J. J., "Influence Lines for Continuous Beams", Oklahoma Engineering Experiment Station Publication, Oklahoma State University, Stillwater, No. 106, 1959.

6. Lassley, T. I., "Beam Constants by High Speed Computer", M. S. Thesis, Oklahoma State University, Stillwater, 1959.

7. Lassley, T. I., French, S. E., and Tuma, J. J., "Volume II, Beam Constants by High Speed Computer", School of Civil Engineering Research Publication, Oklahoma State University, Stillwater, No. 4, 1959.

8. Lassley, T. I., Gauger, F. N., and Tuma, J. J., "Volume III, Influence Lines by High Speed Computer", School of Civil Engineering Research Publication, Oklahoma State University, Stillwater, No. 5, 1960.

9. Oden, J. T., "The Analysis of Fixed-End Frames With Bent Members by the String Polygon Method", M. S. Thesis, Oklahoma State University, Stillwater, 1960.

10. Tuma, J. J., French, S. E., and Lassley, T. I., "Analysis of Continuous Beam Bridges Using Carry-Over Procedure", Oklahoma Engineering Experiment Station Publication, Oklahoma State University, Stillwater, No. 111, 1960.

11. Wu, Chien-Min, "The General String Polygon", M. S. Thesis, Oklahoma State University, Stillwater, 1961.

12. Yu, Hui Sun, "The Analysis of Wedged Frames With Bent Members by the String Polygon Method", M. S. Thesis, Oklahoma State University, Stillwater, 1961.

VITA

Jerry Gene Williams

Candidate for the Degree of

Master of Science

Thesis: ANGULAR FUNCTIONS BY ELECTRONIC COMPUTER

Major Field: Civil Engineering

Biographical:

Personal Date: Born in Ponca City, Oklahoma, July 9, 1939, the son of Glen M. and Faye C. Williams.

Education: Graduated from Central High School, Muskogee, Oklahoma, in May, 1957. Awarded the Association of Oklahoma General Contractors Scholarship in Civil Engineering. Received the Degree of Bachelor of Science in Civil Engineering from Oklahoma State University in May, 1961. Awarded the Continental Oil Company Fellowship in Structural Engineering. Received the Degree of Master of Science from Oklahoma State University with a major in Structural Engineering in August, 1962.

Organizations: Phi Kappa Phi, Omicron Delta Kappa, Chi Epsilon, American Society of Civil Engineers, Pi Mu Epsilon, Who's Who in American Colleges and Universities, International Association for the Exchange of Students for Technical Experience (Exchange student to Norway), Engineer in Training.